



中显液晶
技术资料



中显控制器使用说明书

2009年3月15日

北京市海淀区中关村大街32号和盛大厦811室
电话：(86)-010-52926620 传真：(86)-010-52926621
企业网站：<http://www.zxlcd.com>

点阵字符型液晶显示模块

使用手册

目 录

前言.....	2
第一章 注意事项.....	2
第二章 字符型液晶显示模块的基本特点.....	4
第三章 X 字符型液晶显示模块特性.....	4
第四章 X 字符型液晶显示模块指令集.....	8
第五章 X 字符型液晶显示模块应用.....	11
附录一 HD44780 的内部字符集.....	20

前 言

“ ZX ”北京中显电子有限公司(LCM)标号的代码词头,该系列是目前世界上品种最全的字符型LCM系列,它具有 $8\times 1\sim 40\times 4$ (字符 \times 行)各种规格,广泛应用于智能仪表、通讯、办公自动化及军工等领域。

字符型液晶显示模块由字符型液晶显示屏(LCD),控制驱动主电路HD44780及其扩展驱动电路HD44100或与其兼容的IC,少量阻、容元件,结构件等装配在PCB板上而成。

字符型液晶显示模块目前在国际上已经规范化,无论显示屏规格如何变化,其电特性和接口形式都是统一的。因此只要设计出一种型号的接口电路,在指令设置上稍加改动即可使用各种规格的字符型液晶显示模块。

第一章 注意事项

十分感谢您购买我公司的产品,在使用前请您首先仔细阅读以下注意事项,以免给您造成不必要的损失,您在使用过程中遇到困难时,请拨打我们的技术服务电话,我们将尽力为您提供服务和帮助。

一、处理保护膜

在装好的模块成品表面贴有一层保护膜,以防在装配时沾污显示表面,在整机装配结束前不得揭去,以免弄脏或损坏显示面。

二、加装衬垫

在模块与前面板之间最好加装一块约0.1毫米左右的衬垫。面板还应保持平整,以免在装配后产生扭曲,并可提高其抗振性能。

三、严防静电

模块中的控制、驱动电压是低压、低功耗的CMOS电路,极易被静电击穿,静电击穿是一种不可修复的损坏,而人体有时会产生高达几十伏或上百伏的高压静电,所以,在操作、装配以及使用中都应极其小心,严防静电。为此:

1. 不要用手随意去摸外引线、电路板上的电路及金属框。
2. 如必须直接接触时,应使人体与模块保持在同一电位,或将人体良好接地。
3. 焊接使用的烙铁和操作用的电动工具必须良好接地,没有漏电。
4. 不得使用真空吸尘器进行清洁处理,因为它会产生很强的静电。
5. 空气干燥,也会产生静电,因此,工作间湿度应在RH60%以上。
6. 取出或放回包装袋或移动位置时,也需小心,防止产生静电。不要随意更换包装或舍弃原包装。

四、装配操作时的注意事项

1. 模块是经过精心设计组装而成的,请勿随意自行加工、修整。
2. 金属框爪不得随意扭动、拆卸。
3. 不要随意修改加工PCB板外形、装配孔、线路极其部件。
4. 不得修改导电胶条。
5. 不得修改任何内部支架。

-
6. 不要碰、摔、折曲、扭动模块。

五、焊接

在焊接模块外引线、接口电路时，应按如下规程进行操作。

1. 烙铁头温度小于280℃。
2. 焊接时间小于3~4s。
3. 焊接材料：共晶型、低熔点。
4. 不要使用酸性助焊剂。
5. 重复焊接不要超过3次，且每次重复需间隔5分钟。

六、模块的使用与保养

1. 模块的外引线决不允许接错，在您想调试液晶模块时，请注意正确接线，尤其是正、负电源的接线不能有错，否则可能造成过流、过压、烧电路上的芯片等对液晶模块元器件有损的现象。
2. 模块在使用时，接入电源及断开电源，必须在正电源稳定接入以后，才能输入信号电平。如在电源稳定前或断开后输入信号电平，有可能损坏模块中的IC及电路。
3. 点阵液晶模块显示时的对比度、视角与温度、驱动电压关系很大，所以，如果VEE调整过高，不仅会影响显示，还会缩短模块的使用寿命。
4. 因为液晶材料的物理特性，液晶的对比度会随着温度的变化而相应变化，所以，您加的负电压值应该随温度作相应的调整，大致是温度变化10℃，电压变化1伏。为满足这一要求，您可做一个温度补偿电路，或者安排一个电位器，随温度调整负电压值。
5. 不应在规定工作温度范围以外使用，并且不应在超过存储极限温度的范围外存储，如果温度低于结晶温度，液晶就会结晶，如果温度过高，液晶将变成各向同性的液晶，破坏分子取向，使器件报废。
6. 用力按压显示部分，会产生异常显示。这时切断电源，稍待片刻，重新上电，即恢复正常。
7. 液晶显示器件或模块表面结雾时，不要通电工作，因为这将引起电极化学反应，产生断线。
8. 长期用于阳光及强光下时，被遮部分会产生残留影像。

七、模块的存储

若长期（如几年以上）存储，我们推荐以下方式：

1. 装入聚乙烯口袋（最好有防静电涂层）并将口封住。
2. 在-10℃~ +35℃ 之间存储。
3. 放暗处，避强光。
4. 决不能在表面压放任何物品。
5. 严格避免在极限温/湿度条件下存放。

八、责任范围及维修

在您购买液晶显示模块时，北京中显电子公司将会为您做模块检测，确保您所买的显示模块为完好的器件；在您使用过程中因不小心将显示模块损坏，您可送至蓬远公司维修部修理。液晶模块如果出现屏的问题，比如玻璃面破损、玻璃屏角碎裂等等，将无法进行修理，您的液晶模块只能报废。

公司产品质量都是通过国际质量认证及时间的考验的，请用户放心使用。如果在使用过程中发现问题，可以拿到我公司质量检测部检测，确实是质量问题，而且购买时间在一年之内，我公


司给予换货处理，但是因为用户使用不当（例如静电、焊接、连线不当、过流、过压使用等）所造成的损坏，可送到我们维修部修理。

第二章 字符型液晶显示模块的基本特点

1. 液晶显示屏是以若干个5×8或5×11点阵块组成的显示字符群。每个点阵块为一个字符位，字符间距和行距都为一个点的宽度。
2. 主控制驱动电路为HD44780（HITACHI）及其他公司全兼容电路，如SED1278（SEIKO EPSON）、KS0066（SAMSUNG）、NJU6408（NER JAPAN RADIO）。
3. 具有字符发生器ROM可显示192种字符（160个5×7点阵字符和32个5×10点阵字符，见附录一）。
4. 具有64个字节的自定义字符RAM，可自定义8个5×8点阵字符或4个5×11点阵字符。
5. 具有80个字节的RAM。
6. 标准的接口特性，适配M6800系列MPU的操作时序。
7. 模块结构紧凑、轻巧、装配容易。
8. 单+5V电源供电（宽温型需要一个-7V的驱动电源）。
9. 低功耗、长寿命、高可靠性。

第三章 ZX 字符型液晶显示模块特性

一 ZX 系列型号定义



二 ZX系列接口特性及时序


1. 引脚功能

引脚号	符号	状态	功 能
1	Vss		电源地
2	Vdd		+5V逻辑电源
3	V0		液晶驱动电源
4	RS	输入	寄存器选择 1: 数据; 0: 指令
5	R/W	输入	读、写操作选择 1: 读; 0: 写
6	E	输入	使能信号 (MDLS40466未用, 符号NC)
7	DB0	三态	数据总线 (LSB)
8	DB1	三态	数据总线
9	DB2	三态	数据总线
10	DB3	三态	数据总线
11	DB4	三态	数据总线
12	DB5	三态	数据总线
13	DB6	三态	数据总线
14	DB7	三态	数据总线 (MSB)
*15	E1	输入	MDLS40466上两行使能信号
*16	E2	输入	MDLS40466下两行使能信号

注：15、16两管脚仅用于ZX40466，其余型号不用或为LED背光电源输入。

2. HD44780读、写操作


(1) 写操作 (MPU至HD44780)



Write Operation

项 目	符号	最小值	最大值	单位
使能周期	T _{cycE}	1000	—	ns
使能脉冲宽度	P _{wEH}	450	—	ns
使能升、降时间	T _{er} ,T _{ef}	—	25	ns
地址建立时间	T _{as}	140	—	ns
地址保持时间	T _{ah}	10	—	ns
数据建立时间	T _{dsw}	195	—	ns
数据保持时间	T _h	10	—	ns

(2) 读操作 (HD44780至MPU)



Note: * VOL1 is assumed to be 0.8 V at 2 MHz operation.

Read Operation

项 目	符号	最小值	最大值	单位
使能周期	T _{cycE}	1000	—	ns
使能脉冲宽度	P _{wEH}	450	—	ns
使能升、降时间	T _{er} ,T _{ef}	—	25	ns
地址建立时间	T _{as}	140	—	ns
地址保持时间	T _{ah}	10	—	ns
数据延迟时间	T _{ddr}	—	320	ns
数据保持时间	T _{dhr}	10	—	ns

(3) 信号真值表

RS	R/W	E	功能
0	0	下降沿	写指令代码
0	1	高电平	读忙标志和AC值
1	0	下降沿	写数据
1	1	高电平	读数据

五、 ZX 系列电特性

(1)绝对最大值范围

项目	符号	最小值	最大值	备注
电路逻辑电压(V)	Vdd -Vss	0	7.0	
液晶驱动电压(V)	Vdd -V0	0	13.5	
输入电压(V)	VI	Vss	Vdd	
工作温度(°C)		-5	+50	LV型
		-10	+60	HV型
		-20	+70	HT型
		-30	+80	EHT型
存储温度(°C)		-20	+60	LV型
		-20	+70	HV型
		-30	+80	HT型
		-40	+80	EHT型

(2)电特性

项目	符号	条件	MIN	TYP	MAX	单位
输入高电压	VIH		2.2	-	-	V
输入低电压	VIL		-	-	0.6	V
输出高电压	VOH	IOH = 0.2 mA	2.4	-	-	V
输出低电压	VOL	IOL = 1.2 mA	-	-	0.4	V
电源电流	IDD		-	0.5	0.2	mA

六、 ZX系列供电电路（见附录二）

第四章 X 字符型液晶显示模块指令集

一、指令集

1、清屏

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	0	0	0	1

运行时间 (250Khz) : 1.64 μ s

功能: 清DDRAM和AC值。

2、归位

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	0	0	1	*

运行时间 (250Khz) : 1.64 μ s

功能: AC=0, 光标、画面回HOME位。

3、输入方式设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	0	1	I/D	S

运行时间 (250Khz) : 40 μ s

功能: 设置光标、画面移动方式。

其中: I/D=1: 数据读、写操作后, AC自动增一;

I/D=0: 数据读、写操作后, AC自动减一;

S=1: 数据读、写操作, 画面平移;

S=0: 数据读、写操作, 画面不动。

4、显示开关控制

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	0	1	D	C	B

运行时间 (250Khz) : 40 μ s

功能: 设置显示、光标及闪烁开、关。

其中: D表示显示开关: D=1为开, D=0为关;

C表示光标开关: C=1为开, C=0为关;

B表示闪烁开关: B=1为开, B=0为关。

5、光标、画面位移

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	0	1	S/C	R/L	*	*

运行时间 (250Khz) : 40 μ s

功能: 光标、画面移动, 不影响DDRAM。

其中: S/C=1: 画面平移一个字符位;

S/C=0: 光标平移一个字符位;

R/L=1: 右移; R/L=0: 左移。

6、功能设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	0	1	DL	N	F	*	*

运行时间 (250Khz) : 40 μ s

功能: 工作方式设置 (初始化指令)。

其中: DL=1, 8位数据接口; DL=0, 4位数据接口;

N=1, 两行显示; N=0, 一行显示;

F=1, 5 \times 10点阵字符; F=0, 5 \times 7点阵字符。

7、CGRAM地址设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	0	1	A5	A4	A3	A2	A1	A0

运行时间 (250Khz) : 40 μ s

功能: 设置CGRAM地址。A5~A0=0~3FH。

8、DDRAM地址设置

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	0	1	A6	A5	A4	A3	A2	A1	A0

运行时间 (250Khz) : 40 μ s

功能: 设置DDRAM地址。

N=0, 一行显示A6~A0=0~4FH;

N=1, 两行显示, 首行A6~A0=00H~2FH, 次行A6~A0=40H~67H。

9、读BF及AC值

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
0	1	BF	AC6	AC5	AC4	AC3	AC2	AC1	AC0

功能: 读忙BF值和地址计数器AC值。

其中: BF=1: 忙; BF=0: 准备好。此时, AC值意义为最近一次地址设置 (CGRAM或DDRAM) 定义。

10、写数据

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
1	0	数 据							

运行时间 (250Khz) : 40 μ s

功能: 根据最近设置的地址性质, 数据写入DDRAM或CGRAM内。

11、读数据

RS	R/W	DB7	DB6	DB5	DB4	DB3	DB2	DB1	DB0
1	1	数 据							

运行时间 (250Khz) : 40 μs

功能: 根据最近设置的地址性质, 从DDRAM或CGRAM数据读出。


二、功能设置指令参数N, F值及显示RAM地址对应表

规格	型 号	N F	行	1 2..8 9 10..16 17..20 21..24 25..40
8×1	MDLS81809	0 0	1	00 01..07
16×1	MDLS16163 MDLS16165 MDLS16166 MDLS16168 MDLS161615	1 0	1	00 01..07 40 41..47
16×2	MDLS16265 MDLS16265B MDLS16268 MDLS16268C	1 0	1 2	00 01..07 08 09..0F 40 41..47 48 49..4F
16×4	MDLS16465	1 0	1 2 3 4	00 01..07 08 09..0F 40 41..47 48 49..4F 10 11..17 18 19..1F 50 51..57 58 59..5F
20×1	MDLS20188 MDLS20189	0 0	1	00 01..07 08 09..0F 10..13
20×2	MDLS20265 MDLS20268 MDLS20269	1 0	1 2	00 01..07 08 09..0F 10..13 40 41..47 48 49..4F 50..53
20×4	MDLS20464 MDLS20464B MDLS20468 MDLS204612	1 0	1 2 3 4	00 01..07 08 09..0F 10..13 40 41..47 48 49..4F 50..53 14 15..1B 1C 1D..23 24..27 54 55..5B 5C 5D..63 64..67
24×2	MDLS24265	1 0	1 2	00 01..07 08 09..0F 10..13 14..17 40 41..47 48 49..4F 50..53 54..57
40×2	MDLS40266	1 0	1 2	00 01..07 08 09..0F 10..13 14..17 18..27 40 41..47 48 49..4F 50..53 54..57 58..67
40×4	MDLS40466 注: 这是双片控制器工作, 1、2行和3、4行分别由一片 控制器所控制	1 0	1 2 3 4	00 01..07 08 09..0F 10..13 14..17 18..27 40 41..47 48 49..4F 50..53 54..57 58..67 00 01..07 08 09..0F 10..13 14..17 18..27 40 41..47 48 49..4F 50..53 54..57 58..67

第五章 Z 字符型液晶显示模块应用

一 直接访问方式接口电路及驱动程序

直接访问方式的接口电路是根据精电蓬远公司的演示板电路而提供的。如下图所示。



常温型V0的参考值V0=0.2 v

直接访问方式的驱动子程序如下：

A15 = E的选通信号， A9 = R/W ， A8 = RS

```

COM      EQU    20H      ; 指令寄存器
DAT      EQU    21H      ; 数据寄存器
CW_Add   EQU    8000H    ; 指令口写地址
CR_Add   EQU    8200H    ; 指令口读地址
DW_Add   EQU    8100H    ; 数据口写地址
DR_Add   EQU    8300H    ; 数据口读地址

```

1、读BF和AC值

```

PR0:  PUSH    DPH
      PUSH    DPL
      PUSH    ACC
      MOV     DPTR,#CR_Add    ; 设置指令口读地址
      MOVX   A,@DPTR         ; 读BF和AC值
      MOV    COM,A           ; 存入COM 单元
      POP    ACC
      POP    DPL
      POP    DPH
      RET

```

2、写指令代码子程序

```

PR1:  PUSH    DPH

```

```

        PUSH    DPL
        PUSH    ACC
        MOV     DPTR,#CR_Add    ; 设置指令口读地址

PR11: MOVX    A,@DPTR
        JB     ACC.7,PR11      ; 判BF=0? 是继续
        MOV     A,COM
        MOV     DPTR,#CW_Add    ; 设置指令口写地址
        MOVX   @DPTR,A        ; 写指令代码
        POP     ACC
        POP     DPL
        POP     DPH
        RET

3、写显示数据子程序
PR2:  PUSH    DPH
        PUSH    DPL
        PUSH    ACC
        MOV     DPTR,#CR_Add    ; 设置指令口读地址
PR21: MOVX   A,@DPTR
        JB     ACC.7,PR21      ; 判BF=0? 是继续
        MOV     A,DAT
        MOV     DPTR,#DW_Add    ; 设置数据口写地址
        MOVX   @DPTR,A        ; 写数据
        POP     ACC
        POP     DPL
        POP     DPH
        RET

4、读显示数据子程序
PR3:  PUSH    DPH
        PUSH    DPL
        PUSH    ACC
        MOV     DPTR,#CR_Add    ; 设置指令口读地址

PR31: MOVX   A,@DPTR
        JB     ACC.7,PR31      ; 判BF=0? 是继续
        MOV     DPTR,#DR_Add    ; 设置数据口读地址
        MOVX   A,@DPTR        ; 读数据
        MOV     DAT,A          ; 存入DAT 单元
        POP     ACC
        POP     DPL
        POP     DPH
        RET

```

二 间接控制方式接口电路及驱动程序

间接控制方式(4位总线)是利用HD44780所具有的4位数据总线的功能，简化电路接口的一种接口方式。实用电路如下：



间接控制方式的驱动子程序如下：

RS	EQU	P3.3	；寄存器选择信号
R/W	EQU	P3.4	；读/写选择信号
E	EQU	P3.5	；使能信号

1、读BF和AC值

```

PR0: PUSH ACC
      MOV P1,#0FFH ; P1置位, 准备读
      CLR RS ; RS=0
      SETB R/W ; R/W=1
      SETB E ; E=1
      MOV COM,P1 ; 读BF和AC6-4值
      CLR E ; E=0
      MOV P1,#0FFH ; P1置位, 准备读
      SETB E ; E=1
      MOV A,P1 ; 读AC3-0值
      CLR E ; E=0
      SWAP A ; 转换成 8位数据
      ANL A,#0FH
      ANL COM,#0F0H
      ORL A,COM
      MOV COM,A ; 送入COM 单元
      POP ACC
      RET
    
```

2、写指令代码子程序

```

PR1: PUSH ACC
    
```

	CLR	RS	; RS=0
	SETB	R/W	; R/W=1
PR11:	MOV	P1,#0FFH	; P1置位, 准备读
	SETB	E	; E=1
	MOV	A,P1	; 读BF和AC6-4值
	CLR	E	; E=0
	MOV	C,ACC.7	; BF?进位位C
	SETB	E	; E=1
	CLR	E	; E=0 (读AC3-0值)
	JC	PR11	; 判别BF=1? 是转
	CLR	R/W	; R/W=0
	MOV	P1,COM	; 写入指令代码高 4位
	SETB	E	; E=1
	CLR	E	; E=0
	MOV	A,COM	; 写入指令代码低 4位
	SWAP	A	
	MOV	P1,A	
	SETB	E	; E=1
	CLR	E	; E=0
	POP	ACC	
	RET		

3、写显示数据子程序

PR2:	PUSH	ACC	
	CLR	RS	; RS=0
	SETB	R/W	; R/W=1
PR21:	MOV	P1,#0FFH	; P1置位, 准备读
	SETB	E	; E=1
	MOV	A,P1	; 读BF和AC6-4值
	CLR	E	; E=0
	MOV	C,ACC.7	; BF?进位位C
	SETB	E	; E=1
	CLR	E	; E=0 (读AC3-0值)
	JC	PR21	; 判别BF=1? 是转
	SETB	RS	; RS=1
	CLR	R/W	; R/W=0
	MOV	P1,DAT	; 写入数据高 4位
	SETB	E	; E=1
	CLR	E	; E=0
	MOV	A,DAT	; 写入数据低 4位
	SWAP	A	
	MOV	P1,A	
	SETB	E	; E=1
	CLR	E	; E=0
	POP	ACC	
	RET		

4、读显示数据子程序

```

PR3:  PUSH      ACC
      CLR       RS           ; RS=0
      SETB     R/W         ; R/W=1

PR31: MOV      P1,#0FFH    ; P1置位, 准备读
      SETB     E           ; E=1
      MOV      A,P1        ; 读BF和AC6-4值
      CLR      E           ; E=0
      MOV      C,ACC.7     ; BF?进位位C
      SETB     E           ; E=1
      CLR      E           ; E=0 (读AC3-0值)
      JC       PR31       ; 判别BF=1? 是转
      SETB     RS         ; RS=1
      SETB     R/W        ; R/W=1
      MOV      P1,#0FFH    ; P1置位, 准备读
      SETB     E           ; E=1
      MOV      COM,P1      ; 读数据高 4位
      CLR      E           ; E=0
      MOV      P1,#0FFH    ; P1置位, 准备读
      SETB     E           ; E=1
      MOV      A,P1        ; 读数据低 4位
      CLR      E           ; E=0
      SWAP     A           ; 转换成 8位
      ANL     A,#0FH
      ANL     DAT,#0F0H
      ORL     A,DAT
      MOV      DAT,A       ; 数据送入DAT 单元
      POP     ACC
      RET

```

三、应用程序

该程序是依据液晶显示模块 ZX0802 而编制的。其它规格的字符型液晶显示模块都适用，仅是地址的改动而已。

示例一 初始化子程序

(1) 直接访问方式下的初始化子程序

```

INT:  MOV      A,#30H      ; 此循环必要,不可删
      MOV      DPTR,#CW_Add ; 指令口地址设置
      MOV      R2,#03H    ; 循环量=3

INT1: MOVX     @DPTR,A     ; 写指令代码
      LCALL   DELAY       ; 调延时子程序
      DJNZ   R2,INT1
      MOV     COM,#38H    ; 设置工作方式*
      LCALL   PR1
      MOV     COM,#01H    ; 清屏

```

```

LCALL    PR1
MOV      COM,#06H      ; 设置输入方式
LCALL    PR1
MOV      COM,#0FH      ; 设置显示方式
LCALL    PR1
RET

```

(2) 间接控制方式下的初始化子程序

```

INT:  MOV    P1,#30H      ; 工作方式设置指令代码
      CLR    RS          ; RS=0
      CLR    R/W         ; R/W=0
      MOV    R2,#03H     ; 循环量=3

INT1: SETB   E           ; E=1
      CLR    E           ; E=0
      LCALL  DELAY       ; 调延时子程序
      DJNZ  R2,INT1
      MOV    P1,#28H     ; 设置工作方式*
      SETB  E           ; E=1
      CLR    E           ; E=0
      MOV    COM,#01H    ; 清屏
      LCALL  PR1
      MOV    COM,#06H    ; 设置输入方式
      LCALL  PR1
      MOV    COM,#0FH    ; 设置显示方式
      LCALL  PR1
      RET

```

注：* 工作方式参数根据所使用的字符型液晶显示模块的特性而定。

```

DELAY:  MOV    R6,#00H    ; 延时子程序
      MOV    R7,#00H

```

```

DELAY1: NOP
      DJNZ  R7,DELAY1
      DJNZ  R6,DELAY1
      RET

```

初始化演示程序

```

MAIN:   MOV    SP,#60H
      ANL    P3,#0C7H
      LCALL  INT

```

示例二 字符的写入方式

HD44780有五种字符写入方式。演示程序如下：

1. 逐字依次输入方式演示程序段

```

MOV      COM,#06H
LCALL    PR1

```

```

MOV      COM,#80H
LCALL   PR1
MOV      DPTR,#TAB
MOV      R2,#12
MOV      R3,#00H
WRIN:   MOV      A,R3
        MOVC    A,@A+DPTR
        MOV     DAT,A
        LCALL  PR2
        LCALL  DELAY
        INC   R3
        DINZ  R2,WRIN
        SJMP  $
TAB:    DB  43H,75H,72H,73H,6FH,72H,20H ; Cursor
        DB  57F,72H,69H,74H,65H       ; write

```

2. 光标左移输入方式演示程序段

```

CL_ENTER: MOV      COM,#04H           ; 输入方式设置
          LCALL   PR1
          MOV     COM,#90H           ; 设置DDRAM 地址
          LCALL  PR1
          MOV     DPTR,#TABCL       ; 设置字表首地址
          MOV     R2,#12H           ; 循环量设置

CL_1:    MOV     A,R2
          DEC    A
          MOVC   A,@A+DPTR         ; 取字符代码
          MOV    DAT,A
          LCALL  PR2               ; 写数据
          LCALL  DELAY             ; 演示用延时
          DJNZ   R2,CL_1
          SJMP   $

TABCL:   DB 43H,75H,72H,73H,6FH,72H,20H ; Cursor
          DB 4CH,65H,66H,74H,20H       ; Left
          DB 53H,63H,72H,6FH,6CH,6CH   ; Scroll

```

3. 光标右移输入方式演示程序段

```

CR_ENTER: MOV      COM,#06H           ; 输入方式设置
          LCALL   PR1
          MOV     COM,#80H           ; 设置DDRAM 地址
          LCALL  PR1
          MOV     DPTR,#TABCR       ; 设置字表首地址
          MOV     R2,#13H           ; 循环量设置
          MOV     R3,#00H

CR_1:    MOV     A,R3
          MOVC   A,@A+DPTR         ; 取字符代码
          MOV    DAT,A
          LCALL  PR2               ; 写数据

```

```

INC          R3
LCALL       DELAY          ; 演示用延时
DJNZ       R2,CR_1
SJMP      $

```

```

TABCR:      DB 43H,75H,72H,73H,6FH,72H,20H  ; Cursor
            DB 52H,69H,67H,68H,74H,20H      ; Right
            DB 53H,63H,72H,6FH,6CH,6CH      ; Scroll

```

4. 画面左滚动输入方式演示程序段

```

L_ENTER:    MOV          COM,#07H          ; 输入方式设置
            LCALL       PR1
            MOV         COM,#90H          ; 设置DDRAM 地址
            LCALL       PR1
            MOV         DPTR,#TABL        ; 设置字表首地址
            MOV         R2,#0BH          ; 循环量设置
            MOV         R3,#00H

```

```

L_1:        MOV         A,R3
            MOVC        A,@A+DPTR        ; 取字符代码
            MOV         DAT,A
            LCALL       PR2              ; 写数据
            INC         R3
            LCALL       DELAY            ; 演示用延时
            DJNZ       R2,L_1
            SJMP      $

```

```

TABL:       DB 4CH,65H,66H,74H,20H        ; Left Scroll
            DB 53H,63H,72H,6FH,6CH,6CH

```

5. 画面右滚动输入方式演示程序段

```

R_ENTER:    MOV          COM,#05H          ; 输入方式设置
            LCALL       PR1
            MOV         COM,#80H          ; 设置DDRAM 地址
            LCALL       PR1
            MOV         DPTR,#TABR        ; 设置字表首地址
            MOV         R2,#0CH          ; 循环量设置

```

```

R_1:        MOV         A,R2
            DEC         A
            MOVC        A,@A+DPTR        ; 取字符代码
            MOV         DAT,A
            LCALL       PR2              ; 写数据
            LCALL       DELAY            ; 演示用延时
            DJNZ       R2,R_1
            SJMP      $

```

```

TABR:       DB 52H,69H,67H,68H,74H,20H    ; Right Scroll
            DB 53H,63H,72H,6FH,6CH,6CH

```

示例三 建立自定义字符库

字符型液晶显示模块允许用户在HD44780片内CGRAM内（64个字节00H-3FH）自建立8个5×8点阵的字符。建立自定义字符库的子程序如下：

```
CG_WRITE:  MOV      COM,#40H          ; 设置CGRAM 地址
           LCALL   PR1
           MOV     R2,#64          ; 循环量设置
           MOV     R3,#00H
           MOV     DPTR,#CGTAB    ; 设置字模数据首地址

CG1:       MOV     A,R3
           MOVC   A,@A+DPTR
           MOV     DAT,A
           LCALL  PR2
           INC    R3
           DJNZ   R2,CG1
           RET

CGTAB:     DB 08H,0FH,12H,0FH,0AH,1FH,02H,02H ;"年"代码=00H
           DB 0FH,09H,0FH,09H,0FH,09H,11H,00H ;"月"代码=01H
           DB 1FH,11H,11H,1FH,11H,11H,1FH,00H ;"日"代码=02H
           DB 11H,0AH,04H,1FH,04H,1FH,04H,00H ;"$"代码=03H
           DB 0EH,00H,1FH,0AH,0AH,0AH,13H,00H ;"元"代码=04H
           DB 18H,18H,07H,08H,08H,08H,07H,00H ;"°C"代码=05H
           DB 04H,0AH,15H,04H,04H,04H,04H,00H ;"↑"代码=06H
           DB 17H,15H,15H,15H,15H,15H,17H,00H ;"10"代码=07H
```

自定义字符显示演示程序段

```
           LCALL   CG_WRITE      ; 调建立自定义字符子程序
           MOV     COM,#80H      ; 设置DDRAM 地址
           LCALL   PR1
           MOV     DPTR,#TABDY   ; 设置字表首地址
           MOV     R2,#18        ; 循环量设置
           MOV     R3,#00H

LOOP1:     MOV     A,R3
           MOVC   A,@A+DPTR     ; 取字符代码
           MOV     DAT,A
           LCALL  PR2           ; 写数据
           INC    R3
           DJNZ   R2,LOOP1
           MOV     COM,#0C0H     ; 设置DDRAM 地址
           LCALL  PR1
           MOV     R2,#18        ; 循环量设置

LOOP2:     MOV     A,R3
           MOVC   A,@A+DPTR     ; 取字符代码
           MOV     DAT,A
```

LCALL PR2 ; 写数据
 INC R3
 DJNZ R2,LOOP2
 SJMP \$

TABDY: DB 4DH,44H,4CH,53H,34H,30H,32H,36H,36H
 DB 20H,03H,32H,35H,35H,2EH,30H,30H,04H
 DB 31H,39H,39H,37H,00H,37H,01H,31H,02H
 DB 20H,54H,3DH,33H,35H,05H,06H,20H,07H


附录一：HD44780的内部字符集

	0000	0001	0010	0011	0100	0101	0110	0111	1000	1001	1010	1011	1100	1101	1110	1111
xxxx0000	CG RAM (7)		0	@	P	`	P				-	9	≡	α	p	
xxxx0001	(2)	!	1	A	Q	a	q				。	ア	チ	△	ä	q
xxxx0010	(3)	"	2	B	R	b	r				「	イ	ツ	×	ß	θ
xxxx0011	(4)	#	3	C	S	c	s				」	ウ	テ	モ	ς	∞
xxxx0100	(5)	\$	4	D	T	d	t				√	エ	ト	†	μ	Ω
xxxx0101	(6)	%	5	E	U	e	u				・	オ	ナ	1	ς	Ü
xxxx0110	(7)	&	6	F	V	f	v				ヲ	カ	ニ	ヨ	ρ	Σ
xxxx0111	(8)	'	7	G	W	g	w				ア	キ	ヌ	ラ	g	π
xxxx1000	(1)	(8	H	X	h	x				イ	ク	ネ	リ	J	×
xxxx1001	(2))	9	I	Y	i	y				ウ	ケ	ル		Y	
xxxx1010	(3)	*	:	J	Z	j	z				エ	コ	ハ	レ	j	〒
xxxx1011	(4)	+	;	K	[k	{				オ	サ	ヒ	ロ	*	万
xxxx1100	(5)	,	<	L	¥	l					カ	シ	フ	ワ	¢	円
xxxx1101	(6)	-	=	M]	m	}				ユ	ヌ	ヘ	ン	も	÷
xxxx1110	(7)	.	>	N	^	n	‡				ヨ	セ	ホ	°	ñ	
xxxx1111	(8)	/	?	O	_	o	←				ツ	ソ	マ	°	ö	■


Note: The user can specify any pattern for character-generator RAM.

附录二： X 系列产品供电电路

Normal Temperature Type




Extended Temperature Type



※NOTE: If V_{bk} vary from recommended value, you cannot get proper contrast or viewing angle.

●Examples of Temperature Compensation Circuits for Extended Temp Type. (Only for reference)



Thermistor: $R_{th}(25^{\circ}\text{C})=15[\text{k-ohm}]$, $B=4200[\text{K}]$
 Resistors: $R_p=30[\text{k-ohm}]$, $R_s=6.8[\text{k-ohm}]$, $R_m=3.3[\text{k-ohm}]$
 Transistor: PNP Type
 $V_{cc}: +5\text{V}$, $V_{ss}: 0\text{V}$ (Logic Supply)
 $V_z: -8[\text{V}]$ (-7.8 to $-8.2[\text{V}]$)
 $V_{ee}<V_z[\text{V}]$, $R_z=(V_z-V_{ee})/5[\text{k-ohm}]$

Thermistor: $R_{th}(25^{\circ}\text{C})=15[\text{k-ohm}]$, $B=4200[\text{K}]$
 Resistors: $R_p=510[\text{k-ohm}]$, $R_s=8.2[\text{k-ohm}]$, $R_m=3.9[\text{k-ohm}]$
 Transistor: PNP Type
 $V_{cc}: +5\text{V}$, $V_{ss}: 0\text{V}$ (Logic Supply)
 $V_z: -11[\text{V}]$ (-10.725 to $-11.275[\text{V}]$)
 $V_{ee}<V_z[\text{V}]$, $R_z=(V_z-V_{ee})/5[\text{k-ohm}]$